#region license
# This file is part of Grawlix.
# 
# Grawlix is free software: you can redistribute it and/or modify
# it under the terms of the GNU General Public License as
# published by the Free Software Foundation, either version 3 of
# the License, or (at your option) any later version.
# 
# Grawlix is distributed in the hope that it will be useful,
# but WITHOUT ANY WARRANTY; without even the implied warranty of
# MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE.  See the
# GNU General Public License for more details.
# 
# You should have received a copy of the GNU General Public
# License along with Grawlix.  If not, see
# <http://www.gnu.org/licenses/>.
#endregion

"""The default <IStreamResolver> implementation."""

namespace Grawlix.Services.Resources
import Grawlix.Services
import System

transient class DefaultStreamResolver(ServiceBase, IStreamResolver):
"""The default service that resolves URIs into streams."""
	
	def constructor(uriResolver as IUriResolver):
	"""
	Initializes a new stream resolver.
	Param uriResolver: The URI translator.
	"""
		_uriResolver = uriResolver
	
	def Open(uri as Uri, allowLocal as bool) as IO.Stream:
	"""
	Open a read-only stream from the disk.
	Param uri: The URI which identifies the stream.
	Param allowLocal: Whether to allow local URI schemes,
		such as file:// and internal://.
	Returns: The newly opened stream.
	Raises IOException: A generic I/O error occurred.
	Raises FileNotFoundException: The URI could not be resolved.
	Remarks: Note that this is a blocking operation which should only be
		called asynchronously to avoid making the game hang. Reading from
		the stream is also a blocking operation.
	"""
		stream = TryOpen(uri, allowLocal)
		return stream unless stream is null
		raise IO.FileNotFoundException(
			"A request was made to resolve a universal resource identifier " +
			"(${uri}) to a stream, but no stream was found. " +
			"The stream requested was not necessarily a critical " +
			"component, but the data will not be accessible.")
	
	def TryOpen(uri as Uri, allowLocal as bool) as IO.Stream:
	"""
	Open a read-only stream from the disk.
	Param uri: The URI which identifies the stream.
	Param allowLocal: Whether to allow local URI schemes,
		such as file:// and internal://.
	Returns: The newly opened stream, or null if not found.
	Raises IOException: A generic I/O error occurred.
	Remarks: Note that this is a blocking operation which should only be
		called asynchronously to avoid making the game hang. Reading from
		the stream is also a blocking operation.
	"""
		resolved = _uriResolver.TryResolve(uri, allowLocal)
		return null if resolved is null
		if resolved.Scheme == Uri.UriSchemeFile:
			return IO.FileStream(resolved.LocalPath, IO.FileMode.Open)
		return null
	
	override def ToString():
	"""
	Get a string representation for this object.
	Returns: A string representation for this object.
	"""
		return "DefaultStreamResolver"
	
	private _uriResolver as IUriResolver
	"""The URI translation service."""
